



Министерство науки и высшего образования Российской Федерации
Федеральное государственное автономное образовательное учреждение
высшего образования
«Московский государственный технический университет
имени Н.Э. Баумана
(национальный исследовательский университет)»
(МГТУ им. Н.Э. Баумана)

ФАКУЛЬТЕТ ИНФОРМАТИКА И СИСТЕМЫ УПРАВЛЕНИЯ

КАФЕДРА КОМПЬЮТЕРНЫЕ СИСТЕМЫ И СЕТИ (ИУ6)

НАПРАВЛЕНИЕ ПОДГОТОВКИ 09.03.01 Информатика и вычислительная техника

О Т Ч Е Т

по лабораторной работе № 3

Название: Таймеры микроконтроллеров AVR

Дисциплина: Микропроцессорные системы

Студент

ИУ6-63Б

(Группа)

(Подпись, дата)

М.Ю. Шабалин

(И.О. Фамилия)

Преподаватель

(Подпись, дата)

Е.Ю. Гаврилова

(И.О. Фамилия)

Москва, 2025

Цель работы:

- изучение структуры и режимов работы таймеров,
- программирование таймеров для решения различных задач.

Практическая часть

Исходный код модифицированной программы из задания 1 представлен на листинге 1.

Листинг 1 – Исходный код модифицированной программы

```
.include "m8515def.inc" ;файл определений ATmega8515
.def temp = r16 ;временный регистр
;***Таблица векторов прерываний
.org $000
rjmp INIT ;обработка сброса
.org $007
rjmp T0_OVF ;обработка переполнения таймера T0

;***Инициализация МК
INIT:
    ldi temp,low(RAMEND) ;установка
    out SPL,temp ; указателя стека
    ldi temp,high(RAMEND) ; на последнюю
    out SPH,temp ; ячейку ОЗУ

    clr temp ;инициализация выводов порта PC
    out DDRC,temp ; на ввод
    ldi temp,0x01 ;включение 'подтягивающего' резистора
    out PORTC,temp ; входа PC0

    ser temp ;инициализация выводов порта PB и PD
    out DDRB,temp ; на вывод
    out PORTB,temp
    out DDRD,temp ; на вывод
    out PORTD,temp ;выключение светодиодов

    ldi temp,(1<<SE) ;разрешение перехода
    out MCUCR,temp ; в режим Idle

;***Настройка таймера T0 на режим счётчика событий

    ldi temp,0x02 ;разрешение прерывания по
    out TIMSK,temp ; переполнению таймера T0

    ldi temp,0x07 ;переключение таймера
    out TCCR0,temp ; по положительному перепаду напряжения

    sei ;глобальное разрешение прерываний

    ldi temp,0xFC ;$FC=-4 для
    out TCNT0,temp ; отсчёта 4-х нажатий
```

Продолжение листинга 1

```
LOOP: sbic PINC, 0
      rjmp LOOP
;положительный перепад напряжения (0/1) на PB0
      cbi PORTB,0
      sbi PORTB,0
      rcall DELAY
wait:
      sbis PINC, 0
      rjmp wait
      rjmp LOOP
T0_OVF:
      clr temp
      out PORTD,temp ;включение светодиодов
      rcall DELAY ;задержка
      ser temp
      out PORTD,temp ;выключение светодиодов
      ldi temp,0xFC ;перезагрузка
      out TCNT0,temp ; TCNT0
      reti
DELAY:
      ldi r19,6
      ldi r20,255
      ldi r21,255
dd: dec r21
     brne dd
     dec r20
     brne dd
     dec r19
     brne dd
     ret
```

Исходный код модифицированной программы из задания 2 представлен на листинге 2. Расчеты приведены ниже:

$$t = (65536 - TCNT1) \frac{K}{F_{clk}} = 400 \text{ мс},$$

где $K = 64$ – коэффициент предделителя рабочей частоты микроконтроллера $F_{clk} = 3,69 \text{ МГц}$. Следовательно, начальное значение таймера:

$$TCNT1 = 65536 - \frac{tF_{clk}}{K} = 65536 - \frac{0,4 \times 3,69 \times 10^6}{64} = 42473,5 < 42474.$$

Листинг 2 – Исходный код модифицированной программы

```
.include "m8515def.inc" ;файл определений для ATmega8515
.def temp1 = r16 ;временный регистр 1
.def temp2 = r17 ;временный регистр 2
.def reg_led = r20
.def reg_led_out = r24
.def temp = r19
.def left = r21
.def right = r22
```

Продолжение листинга 2

```
.equ led = 0      ;0-й бит порта PB
.equ start = 0    ;0-й бит порта PA
.equ stop = 1     ;1-й бит порта PA

.macro reload_timer ;макрос
    ldi temp1,high(55000) ; для загрузки в
    ldi temp2,low(55000)  ; таймер
    out TCNT1H,temp1      ; исходного значения
    out TCNT1L,temp2      ; для начала отсчета
.endmacro

;***Таблица векторов прерываний***
.org $0000
    rjmp INIT ; обработка сброса
.org $0006
    rjmp T1_OVF ; обработка переполнения таймера

;***Инициализация МК***
INIT:
    ldi reg_led,0x18
    ldi left, 0x10
    ldi right, 0x08

    ldi temp1,high(RAMEND) ;установка
    out SPH,temp1 ; указателя стека
    ldi temp1,low(RAMEND) ; на последнюю
    out SPL,temp1 ; ячейку ОП

    ser temp1 ;инициализация выводов
    out DDRB,temp1 ; порта PB на вывод
    out PORTB,temp1 ;погасить LED

    clr temp1 ;инициализация
    out DDRA,temp1 ; порта PA на ввод
    ldi temp1,0b00000011 ;включение подтягивающих
    out PORTA,temp1 ; резисторов порта PA

    reload_timer ;загрузка в таймер исходного значения

    ldi temp1,(1<<TOIE1) ;разрешение прерывания
    out TIMSK,temp1 ; таймера по переполнению

    set ;установка флага T=1
    sei ;глобальное разрешение прерываний

test_start:
    sbic PINA,start ;проверка состояния кнопки start
    rjmp test_stop
    ldi temp1,((1<<CS11)|(1<<CS10))
    out TCCR1B,temp1 ; включение таймера с делителем
wait_0:
    sbis PINA,start ;проверка отпускания кнопки
```

Продолжение листинга 2

```

rjmp wait_0
test_stop:
    sbic PINA, stop ;проверка состояния кнопки stop
    rjmp test_start
    clr temp1
    out TCCR1B, temp1 ;выключение таймера
wait_1:
    sbis PINA, stop ;проверка отпускания кнопки
    rjmp wait_1
    rjmp test_start

;***Обработка прерывания при переполнении таймера T1
T1_OVF:
    clr temp1
    out TCCR1B, temp1 ;отключение таймера
    brts switch_on ;проверка значения в флаге T

    set ;установка флага T=1
    rjmp set_timer
switch_on:
    ser reg_led_out
    eor reg_led_out, reg_led
    out PORTB, reg_led_out

    lsl left
    lsr right
    brcc SKIP_SETUP
    ldi left, 0x01
    ldi right, 0x80
SKIP_SETUP:
    ldi reg_led, 0x00
    or reg_led, left
    or reg_led, right
    clt ;сброс флага T=0
set_timer:
    reload_timer ;загрузка в таймер исходного значения
    ldi temp1, ((1<<CS11)|(1<<CS10))
    out TCCR1B, temp1 ; включение таймера
    reti

```

Результаты 3-х замеров временных интервалов при помощи платы STK500 и секундомера и оценка погрешности представлены в таблице 1.

Таблица 1 – Замеры и оценка погрешности

H byte	L byte	Расчет	Real Time	Отн. погрешность
00011111	10000011	8067 = 2.2 с	2с 19мс	0.004
00111111	01010010	16210 = 4.49 с	4с 39мс	0.02
01100110	01111111	26239 = 7.28 с	7с 29мс	0.001

Модифицированная программа для задания 4 представлена на листинге 3. Временная диаграмма для этого задания представлена на рисунке 1. Расчеты приведены ниже:

$$f_{oco} = \frac{F_{clk}}{2 * K * (1 + OCR0)} = 261.63 \text{ Гц, где } OCR0 = 109$$

Листинг 3 – Модифицированная программа для задания 4

```
.include "m8515def.inc"
.def temp = r16
.equ START = 0
.equ STOP = 1
.equ PRESCALER_64 = ((1<<CS01)|(1<<CS00))
.equ TIMER_SETTINGS = ((1<<WGM01)|(1<<COM00)|PRESCALER_64)
.equ VALUE = 109 ; OCR0, нота «ДО» первой октавы

.org $000
rjmp INIT

INIT:
ldi temp,high(RAMEND)
out SPH,temp
ldi temp,low(RAMEND)
out SPL,temp

ser temp
out DDRB,temp
out PORTB,temp

clr temp
out DDRA,temp
ldi temp,0b00000011
out PORTA,temp

ldi temp,VALUE
out OCR0,temp
test_start:
sbic PINA,START
rjmp test_stop
ldi temp,TIMER_SETTINGS
out TCCR0,temp

wait_0:
sbis PINA,START
rjmp wait_0

test_stop:
sbic PINA,STOP
rjmp test_start
```

Продолжение листинга 3

```
clr temp
out TCCR0,temp

wait_1:
sbis PINA,STOP
rjmp wait_1
rjmp test_start
```



Рисунок 1 – Временная диаграмма, задание 4

Модифицированная программа для задания 5 представлена на листинге 4. Временная диаграмма для этого задания представлена на рисунке 2. Расчеты приведены ниже:

$$f_{OC1A} = \frac{F_{clk}}{2 * K * TOP} = \frac{3.69}{2 * 64 * 255} = 7.24 \text{ кГц}$$

Для установки яркости светодиода нужно использовать формулу

$$k_{\text{зап}} = \frac{OCR1A}{TOP}, \text{ где } k = 0.25 \Rightarrow OCR1A = 64$$

Листинг 4 – Модифицированная программа для задания 5

```
.include "m8515def.inc"
.def temp = r16
.equ led = 0
.equ start = 0
.equ stop = 1
.equ PRESCALER_1 = (1<<CS10)
.equ TIMER_SETTINGS = ((1<<COM1A1) | (1<<COM1A0) | (1<<WGM10))
.equ VALUE = 64

.macro pwm_on
ldi temp,TIMER_SETTINGS
```

Продолжение листинга 4

```
    out TCCR1A,temp
    ldi temp,PRESALER_1
    out TCCR1B,temp
.endmacro

.macro pwm_off
    clr temp
    out TCCR1A,temp
    out TCCR1B,temp
.endmacro

.org $000
rjmp INIT

INIT:
    ldi temp,high(RAMEND)
    out SPH,temp
    ldi temp,low(RAMEND)
    out SPL,temp

    ser temp
    out DDRD,temp
    out PORTD,temp

    clr temp
    out DDRA,temp
    ldi temp,0b00000011
    out PORTA,temp

    ldi temp,high(VALUE)
    out OCR1AH,temp

    ldi temp,low(VALUE)
    out OCR1AL,temp
test_on:
    sbic PINA,start
    rjmp test_off
    pwm_on
wait_0:
    sbis PINA,start
    rjmp wait_0
test_off:
    sbic PINA,stop
    rjmp test_on
    pwm_off

    ser temp
    out PORTD,temp
wait_1:
    sbis PINA,stop
    rjmp wait_1
    rjmp test_on
```




Рисунок 2 – Временная диаграмма, задание 5

Выводы: в ходе выполнения лабораторной работы были изучены структуры и режимы работы таймеров, а также приобретены навыки программирования таймеров для решения различных задач.